

Paper 077-2011

Create An Excel Pivot Table Using DDE

Kevin Chung, Fannie Mae, Washington DC

ABSTRACT

Pivot Table is one of the practical and prevailing features in Excel. It can be used to summarize, analyze, and investigate the raw data in an effective and efficient fashion. Several presentations with respect to this topic by using VB scripts method have been introduced at SAS® conference in the past couple years. This paper provides an alternative approach, SAS filename with DDE method as well as Excel PIVOT functions, to create a custom formatted Pivot Table in Excel application. Unfortunately, DDE is only available on Windows environment. The author also provides an easy way to transfer SAS data set from Unix to Windows platform.

INTRODUCTION

The key components used in this paper are three major PIVOT functions which can be used to create an empty Pivot Table, add SAS class fields to construct page, row, and column dimensions, and add data fields to compute statistics such as SUM, COUNT, and AVERAGE. A SAS-supplied data set, sashelp.prdsale, is used to illustrate the creation of a pivot table. To demonstrate the data transfer from Unix, the same data set was copied to /tmp directory on Unix as a test data. The source codes and outputs are also included in the Appendix for reference. Before we walkthrough the detailed SAS codes, we can take a look at three major PIVOT functions, PIVOT.TABLE.WIZARD, PIVOT.ADD.FIELDS, and PIVOT.ADD.DATA and see how they perform the creation of a pivot table. For each functions, only those arguments used in the codes will be illustrated. Please refer to References [2] for more information. SAS V9.1.3, PuTTY, and Excel 2003 with SP3 are used in this paper.

Test Data

The SAS-supplied data set sashelp.prdsale is used as the input data. The table below is the partial lists of the input data.

	A	B	C	D	E	F	G	H	I	J
1	ACTUAL	PREDICT	COUNTRY	REGION	DIVISION	PRODTYPE	PRODUCT	QUARTER	YEAR	MONTH
2	\$925.00	\$850.00	CANADA	EAST	EDUCATION	FURNITURE	SOFA	1	1993	Jan
3	\$999.00	\$297.00	CANADA	EAST	EDUCATION	FURNITURE	SOFA	1	1993	Feb
4	\$608.00	\$846.00	CANADA	EAST	EDUCATION	FURNITURE	SOFA	1	1993	Mar
5	\$642.00	\$533.00	CANADA	EAST	EDUCATION	FURNITURE	SOFA	2	1993	Apr
6	\$656.00	\$646.00	CANADA	EAST	EDUCATION	FURNITURE	SOFA	2	1993	May
7	\$948.00	\$486.00	CANADA	EAST	EDUCATION	FURNITURE	SOFA	2	1993	Jun
8	\$612.00	\$717.00	CANADA	EAST	EDUCATION	FURNITURE	SOFA	3	1993	Jul
9	\$114.00	\$564.00	CANADA	EAST	EDUCATION	FURNITURE	SOFA	3	1993	Aug
10	\$685.00	\$230.00	CANADA	EAST	EDUCATION	FURNITURE	SOFA	3	1993	Sep
11	\$657.00	\$494.00	CANADA	EAST	EDUCATION	FURNITURE	SOFA	4	1993	Oct
12	\$608.00	\$903.00	CANADA	EAST	EDUCATION	FURNITURE	SOFA	4	1993	Nov
13	\$353.00	\$266.00	CANADA	EAST	EDUCATION	FURNITURE	SOFA	4	1993	Dec
14	\$107.00	\$190.00	CANADA	EAST	EDUCATION	FURNITURE	SOFA	1	1994	Jan
15	\$354.00	\$139.00	CANADA	EAST	EDUCATION	FURNITURE	SOFA	1	1994	Feb
16	\$101.00	\$217.00	CANADA	EAST	EDUCATION	FURNITURE	SOFA	1	1994	Mar
17	\$553.00	\$560.00	CANADA	EAST	EDUCATION	FURNITURE	SOFA	2	1994	Apr
18	\$877.00	\$148.00	CANADA	EAST	EDUCATION	FURNITURE	SOFA	2	1994	May
19	\$431.00	\$762.00	CANADA	EAST	EDUCATION	FURNITURE	SOFA	2	1994	Jun
20	\$511.00	\$457.00	CANADA	EAST	EDUCATION	FURNITURE	SOFA	3	1994	Jul

Table 1

Transfer Data from Unix

This part can be ignored if the input data is created under Windows platform. The purpose of this step is to demonstrate how to move data from Unix to Windows. A third party utility, pscp (PuTTY Secure Copy client) is used to transfer data from Unix to Windows platform. Please refer to References [3] to download this free utility. The contents of pass.inc in Appendix A contain three macro variables which define username/password and Unix server name.

```
%let user=xxxxxxx;
%let pass=xxxxxxx;
%let server=xxxxxxx;
```

The SAS X statement is used to call pscp to copy data from Unix to &dir folder.

```
%inc "&dir\pass.inc"; /* contain &user, &pass, and &dir */
x "pscp -p -pw &pass &user@&server:/tmp/prdsale.sas7bdat &dir";
```

Two command line options were used:

```
-p      preserve file attributes
-pw     Unix server password
```

Populate Data On Excel Worksheet

To populate prdsale data on the worksheet, the filename with DDE method is used below. Three user-defined macros, %nobs, %nfields, and %var_list, are used to generate number of observation, number of fields, and variable name list from prdsale data set.

```
%let nobs=%nobs(lib.prdsale);
%let nfields=%nfields(lib.prdsale);
%let var_list=%var_list(lib.prdsale);
%let var_tab_list="%sysfunc(tranwrd(&var_list,%str( ), "%str( )&tab%str( )"))";
%let range= r1c1:r%eval(&nobs+1)c%&nfields;

/* populate data */
filename out DDE "excel|&dir\[&xls_name]data!&range" notab;
data _null_;
  file out dsd dlm=&tab;
  set lib.prdsale;
  if (_n_=1) then put &var_tab_list;
  put &var_list;
run;
```

Macro variable &var_list holds all the variables based on their position in data set prdsale.

```
ACTUAL PREDICT COUNTRY REGION DIVISION PRODTYPE PRODUCT QUARTER YEAR MONTH
```

The tranwrd function is to replace each blank between ACTUAL and MONTH in &var_list with " '09'x " so that the new macro variable &var_tab_list can be applied by PUT statement to write the header at the first row. The header can be used by PIVOT functions to construct the PivotTable.

```
"ACTUAL" '09'x "PREDICT" '09'x "COUNTRY" . . . . "YEAR" '09'x "MONTH"
```

Three major components introduced below are used to create a pivot table.

Create an empty Pivot Table

The first PIVOT function is PIVOT.TABLE.WIZARD which can be used to create an empty pivot table. It's equivalent to select the "PivotTable and PivotChart Report" from Data menu in Excel 2003 application (Table 2).




Table 2



Table 3

The Source field in PIVOT.TABLE.WIZARD is equivalent to select data worksheet, highlight cells A1 to J1441 (there are 1440 observations in prdsale). The descriptions of each argument in PIVOT.TABLE.WIZARD are listed below. Please refer to Reference [2] for more information.

PIVOT.TABLE.WIZARD (type, source, destination, name, row_grand, col_grand, save_data, apply_auto_format, autopage)

Codes used in paper

PIVOT.TABLE.WIZARD (1, "data!r1c1:r&end_row.c&nfields", "&sheet!r1c1", "PivotTable1", TRUE, TRUE, TRUE, TRUE,)

type a number specifying the type of source data used to create the PivotTable.

Value	Type of source data
1	Microsoft Excel List or Database
2	External data source
3	Multiple consolidation ranges

Source 1 – a cell reference or name to the range to be used as the PivotTable source.
 2 – a one- dimensional array describing the external database to be used as the PivotTable source.
 3 – a multi-dimensional array listing the cell ranges and associated page field items describing the consolidation PivotTable source.

Destination is a cell reference or name. The upper left corner of this range will act as the upper left corner of the PivotTable which will be created. If destination is omitted, Microsoft Excel will choose a default location for the PivotTable.

Name	is the name of the PivotTable to be created given as a text. If name is omitted, Microsoft Excel will choose a default name.
Row_grand	is a logical value which if TRUE displays a Grand Total for each row on the PivotTable. If FALSE, a Grand Total for each row is not displayed.
Col_grand	is a logical value which if TRUE displays a Grand Total for each column. If FALSE, a Grand Total for each column is not displayed.
Save_data	is a logical value which if TRUE causes the data for the PivotTable to be saved along with the PivotTable definition. If FALSE, the data is not saved along with the PivotTable definition.
Apply_auto_format	is a logical value which if TRUE causes autoformatting upon pivoting or refreshing. If FALSE, the PivotTable will not be formatted automatically upon pivoting or refreshing.

Add fields to a PivotTable

The second component is to define page, row, and column dimensions in a PivotTable. The PIVOT.ADD.FIELDS function is to add SAS variables to construct the page, row, and column dimensions. This is equivalent to Table 4 below if the Finish button is clicked on PivotTable and PivotChart Wizard – Step 1 of 3 (Table 3) and the fields are manually dragged and dropped to each area.




Table 4

The descriptions of each arguments used in the functions are listed below.

PIVOT.ADD.FIELDS(name, row_array, column_array, page_array, add_to_table)

[Codes used in paper](#)

PIVOT.ADD.FIELDS("PivotTable1", "Country", "product", "year")

Name	is the name of the PivotTable to which the user wants to add fields. If name is omitted, Microsoft Excel will use the PivotTable containing the active cell.
------	--

Row_array is an array of text constants consisting of the names of the fields which the user would like to add to the PivotTable as Row Fields.

Column_array is an array of text constants consisting of the names of the fields which the user would like to add to the PivotTable as Column Fields.

Page_array is an array of text constants consisting of the names of the fields which the user would like to add to the PivotTable as Page Fields.

The diagram below shows the area for each dimensions.

	A	B	C	D	E	F	G
1	YEAR	(All)					
2							
3	Sum of ACTUAL	PRODUCT					
4	COUNTRY	BED	CHAIR	DESK	SOFA	TABLE	Grand Total
5	CANADA	47,729	50,239	52,187	50,135	46,700	246,990
6	GERMANY	46,134	47,105	48,502	55,060	49,197	245,998
7	U.S.A.	48,174	50,936	48,543	43,393	46,303	237,349
8	Grand Total	142,037	148,280	149,232	148,588	142,200	730,337

Table 5

A variation is shown below. There is no page dimension. Two fields are used in column area.

Codes used in paper

```
PIVOT.ADD.FIELDS("PivotTable2a", "product", {"year", "Country"})
```

Sum of PREDICT	YEAR	COUNTRY							
	1993								
				1993 Total		1994		1994 Total	Grand Total
PRODUCT	CANADA	GERMANY	U.S.A.			CANADA	GERMANY	U.S.A.	
BED	22,640	23,040	25,243	70,923		21,575	20,756	24,613	66,944
CHAIR	24,008	22,991	22,499	69,498		22,788	21,078	22,746	66,612
DESK	23,408	19,821	25,598	68,827		25,985	24,818	26,565	77,368
SOFA	24,515	27,040	24,923	76,478		21,211	22,477	20,285	63,973
TABLE	24,758	24,227	25,500	74,485		22,131	25,306	23,750	71,187
Grand Total	119,329	117,119	123,763	360,211		113,690	114,435	117,959	346,084

Add Data fields to a PivotTable

The last component used to construct a pivot table is to add the data fields to the DATA area. In this example, only one field, ACTUAL, is added with two different functions, Average and SUM.

PIVOT.ADD.DATA(name, pivot_field_name, new_name, position, function, calculation, base_field, base_item, format_text)

Codes used in paper

```
PIVOT.ADD.DATA("PivotTable3", "actual", , 1, 8, , , , "0.00")
```

```
PIVOT.ADD.DATA("PivotTable3", "actual", , 2, , , , "###,##0")
```

2: Sum
4: Count
8: Average

name

position

23	YEAR	(All)						
24								
25			PRODUCT					
26	COUNTRY	Data	BED	CHAIR	DESK	SOFA	TABLE	Grand Total
27	CANADA	Average of ACTUAL	497.18	523.32	543.61	522.24	486.46	514.56
28		Sum of ACTUAL	47,729	50,239	52,187	50,135	46,700	246,990
29	GERMANY	Average of ACTUAL	480.56	490.68	505.23	573.54	512.47	512.50
30		Sum of ACTUAL	46,134	47,105	48,502	55,060	49,197	245,998
31	U.S.A.	Average of ACTUAL	501.81	530.58	505.66	452.01	482.32	494.48
32		Sum of ACTUAL	48,174	50,936	48,543	43,393	46,303	237,349
33	Total Average of ACTUAL		493.18	514.86	518.17	515.93	493.75	507.18
34	Total Sum of ACTUAL		142,037	148,280	149,232	148,588	142,200	730,337

Table 6

Name is the name of the PivotTable to which the user wants to add as a data field. If name is omitted, Microsoft Excel will use the PivotTable containing the active cell.

Pivot_field_name is the name of a field which the user would like to add to his PivotTable as data or as text.

New_name is the name you would like to give to the new field once it is added to your PivotTable. If this argument is omitted, Microsoft Excel will pick a default name for you. This function returns new_name or the name Microsoft Excel chooses for the field.

Position is the position within all the Data fields you would like to place the new data field. If position is omitted, the field will be added as the last data field.

Function is a number from 2 to 2048 specifying how the new field is to be calculated. To compute the value to place in this column choose one value from the following table. If function is omitted, SUM will be used. If the field is a numeric field or text field, COUNTA will be used.

Execute the Pivot Functions

The last step is to create the Pivot functions illustrated above in the MS Excel 4.0 Macro sheet and execute the macros. The source codes listed in Appendix A generates the macros listed in Appendix B. The last data step in source code creates four pivot tables on the same worksheet. The pivot table C in the Sales worksheet in Appendix B only displays the partial output due to the width of the screenshot.

Appendix A

```

/* pivot_sales.sas */

option noxwait;

%let dir=D:\SGF2011;
%let xls_name=Sales.xls;
%let sheet=Sales;
%let tab='09'x;
libname lib "&dir";

options sasautos=("&dir" sasautos);

/*
%inc "&dir\pass.inc";
x "pscp -p -pw &pass &user@&server:/tmp/prdsale.sas7bdat &dir";
*/

%let nobs=%nobs(lib.prdsale);
%let nfields=%nfields(lib.prdsale);
%let var_list=%var_list(lib.prdsale);
%let var_tab_list="%sysfunc(tranwrd(&var_list,%str( ),"%str( )&tab%str( )"))";
%let range=r1c1:r%eval(&nobs+1)c&nfields;
%put nobs=&nobs nfields=&nfields;
%put var_list=&var_list;
%put var_tab_list=&var_tab_list;
%put range=&range;

/* populate data */
filename out DDE "excel|&dir\[&xls_name]data!&range" notab;
data _null_;
  file out dsd dlm=&tab;
  set lib.prdsale;
  if (_n_=1) then put &var_tab_list;
  put &var_list;
run;

filename xlmacro DDE "excel|&dir\[&xls_name]Macro1!r1c1:r150c1" notab;
data _null_;
  file xlmacro;
  put
    "=WORKBOOK.SELECT( ""data"" )"/
    "=ERROR(FALSE)"/
    "=SELECT( ""&range"" )"/
    "=PIVOT.TABLE.WIZARD(1, ""data!&range"", ""&sheet!r1c1"", ""PivotTable1"",
      TRUE,TRUE,TRUE,TRUE, )"/
    "=PIVOT.ADD.FIELDS( ""PivotTable1"", ""Country"", ""product"", ""year"" )"/
    "=PIVOT.ADD.DATA( ""PivotTable1"", ""actual"", , , , , , ""###,##0"" )"/
    "=WORKBOOK.SELECT( ""data"" )"/
    "=SELECT( ""&range"" )"/
    "=PIVOT.TABLE.WIZARD(1, ""data!&range"", ""&sheet!r13c1"", ""PivotTable2"",
      TRUE,TRUE,TRUE,TRUE, )"/
    "=PIVOT.ADD.FIELDS( ""PivotTable2"", ""product"", ""Country"", ""year"" )"/
    "=PIVOT.ADD.DATA( ""PivotTable2"", ""predict"", , , , , , ""###,##0"" )"/
    "=WORKBOOK.SELECT( ""data"" )"/
    "=SELECT( ""&range"" )"/
    "=PIVOT.TABLE.WIZARD(1, ""data!&range"", ""&sheet!r13c7"", ""PivotTable2a"",
      TRUE,TRUE,TRUE,TRUE, )"/

```

```

    "=PIVOT.ADD.FIELDS("PivotTable2a","product",{ "year","Country"})"/
    "=PIVOT.ADD.DATA("PivotTable2a","predict",,,,,,"###,##0")"/
    "=WORKBOOK.SELECT("data")"/
    "=SELECT("&range")"/
    "=PIVOT.TABLE.WIZARD(1,"&range","&sheet!r25c1","PivotTable3",
        TRUE,TRUE,TRUE,TRUE,)/
    "=PIVOT.ADD.FIELDS("PivotTable3","Country","product","year")"/
    "=PIVOT.ADD.DATA("PivotTable3","actual",,1,8,,,,"0.00")"/
    "=PIVOT.ADD.DATA("PivotTable3","actual",,2,,,,"###,##0")"/
    "=HALT(TRUE)"/
    "!dde_flush";
run;

/* Execute macro */
filename cmds DDE 'excel|system' lrecl=200;
data _null_;
    file cmds;
    put "[WORKBOOK.ACTIVATE("&xls_name")]"/
        '[error(false)]'/* Suppress the prompt */
        '[run("Macro1!r1c1")]'
        "[save.as("&dir\&xls_name",1)]";
run;

```


Appendix B

Macro1 Worksheet

	A
1	=WORKBOOK.SELECT("data")
2	=ERROR(FALSE)
3	=SELECT("r1c1:r1441c10")
4	=PIVOT.TABLE.WIZARD(1,"data!r1c1:r1441c10","Sales!r1c1","PivotTable1",TRUE,TRUE,TRUE,TRUE,)
5	=PIVOT.ADD.FIELDS("PivotTable1","Country","product","year")
6	=PIVOT.ADD.DATA("PivotTable1","actual",,,,,,"###,##0")
7	=WORKBOOK.SELECT("data")
8	=SELECT("r1c1:r1441c10")
9	=PIVOT.TABLE.WIZARD(1,"data!r1c1:r1441c10","Sales!r13c1","PivotTable2",TRUE,TRUE,TRUE,TRUE,)
10	=PIVOT.ADD.FIELDS("PivotTable2","product","Country","year")
11	=PIVOT.ADD.DATA("PivotTable2","predict",,,,,,"###,##0")
12	=WORKBOOK.SELECT("data")
13	=SELECT("r1c1:r1441c10")
14	=PIVOT.TABLE.WIZARD(1,"data!r1c1:r1441c10","Sales!r13c7","PivotTable2a",TRUE,TRUE,TRUE,TRUE,)
15	=PIVOT.ADD.FIELDS("PivotTable2a","product",{"year","Country"})
16	=PIVOT.ADD.DATA("PivotTable2a","predict",,,,,,"###,##0")
17	=WORKBOOK.SELECT("data")
18	=SELECT("r1c1:r1441c10")
19	=PIVOT.TABLE.WIZARD(1,"r1c1:r1441c10","Sales!r25c1","PivotTable3",TRUE,TRUE,TRUE,TRUE,)
20	=PIVOT.ADD.FIELDS("PivotTable3","Country","product","year")
21	=PIVOT.ADD.DATA("PivotTable3","actual",,1,8,,,"0.00")
22	=PIVOT.ADD.DATA("PivotTable3","actual",,2,,,"###,##0")
23	=HALT(TRUE)

Sales worksheet

	A	B	C	D	E	F	G	H	I
1	YEAR	(All)							
2									
3	Sum of ACTUAL	PRODUCT							
4	COUNTRY	BED	CHAIR	DESK	SOFA	TABLE	Grand Total		
5	CANADA	47,729	50,239	52,187	50,135	46,700	246,990		
6	GERMANY	46,134	47,105	48,502	55,060	49,197	245,998		
7	U.S.A.	48,174	50,936	48,543	43,393	46,303	237,349		
8	Grand Total	142,037	148,280	149,232	148,588	142,200	730,337		
9									
10									
11	YEAR	(All)							
12									
13	Sum of PREDICT	COUNTRY							
14	PRODUCT	CANADA	GERMANY	U.S.A.	Grand Total				
15	BED	44,215	43,796	49,856	137,867				
16	CHAIR	46,796	44,069	45,245	136,110				
17	DESK	49,393	44,639	52,163	146,195				
18	SOFA	45,726	49,517	45,208	140,451				
19	TABLE	46,889	49,533	49,250	145,672				
20	Grand Total	233,019	231,554	241,722	706,295				
21									
22									
23	YEAR	(All)							
24									
25			PRODUCT						
26	COUNTRY	Data	BED	CHAIR	DESK	SOFA	TABLE	Grand Total	
27	CANADA	Average of ACTUAL	497.18	523.32	543.61	522.24	486.46	514.56	
28		Sum of ACTUAL	47,729	50,239	52,187	50,135	46,700	246,990	
29	GERMANY	Average of ACTUAL	480.56	490.68	505.23	573.54	512.47	512.50	
30		Sum of ACTUAL	46,134	47,105	48,502	55,060	49,197	245,998	
31	U.S.A.	Average of ACTUAL	501.81	530.58	505.66	452.01	482.32	494.48	
32		Sum of ACTUAL	48,174	50,936	48,543	43,393	46,303	237,349	
33	Total Average of ACTUAL		493.18	514.86	518.17	515.93	493.75	507.18	
34	Total Sum of ACTUAL		142,037	148,280	149,232	148,588	142,200	730,337	

CONCLUSION

This paper presents a handy tool to create the PivotTable on Excel using DDE method. Four examples are demonstrated above based on different scenarios. For illustration reason, several values are hard-coded; but you can change them easily by using macro techniques. The advantage of this tip is that you can easily replace one of the four examples with your data elements and create a Pivot table quickly. So give it a try and you can find it's a very useful tool in your daily routine job.

REFERENCES

- [1] SAS OnlineDoc® 9.1.3, SAS Institute Inc. Cary, NC.
<http://support.sas.com/onlinedoc/913/docMainpage.jsp>
- [2] macrofun.hlp
Can be downloaded from <http://support.microsoft.com/kb/128185>
- [3] PuTTY
Can be downloaded from <http://www.putty.org/>

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Feel free to contact the author at:

Kevin Chung
Fannie Mae
4000 Wisconsin Ave., NW
Mail Stop: 2H-4S/07
Washington, DC 20016
Work Phone: 202-752-1568
E-mail: kevin_chung@fanniemae.com
kchung01@hotmail.com

You can download all source codes and presentation slides from www.kevin-chung.com/SGF2011

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.
Other brand and product names are trademarks of their respective companies.